

Exponential Models for Practice Effects in Number Match 3 Response Times

The models for this example come from Hoffman (in preparation) chapter 6. We will be examining change in response time (RT) in milliseconds over six practice sessions to a measure of processing speed (as measured by the number match 3 test) in a sample of 101 older adults. Previously we used polynomial and piecewise slopes models to describe change in RT by session; now we use an exponential model instead, which will require learning a new procedure—SAS PROC NLMIXED. REML is not available in NLMIXED, so these models will be estimated using ML instead. Additional options related to estimation are specified below. To illustrate NLMIXED, we begin with two familiar models: an empty means, random intercept model (1b), and a random quadratic time model (3b).

Model 1b. Empty Means, Random Intercept Model via MIXED and NLMIXED

Level 1: $y_{ti} = \beta_{0i} + e_{ti}$ Level 2: Intercept: $\beta_{0i} = \gamma_{00}$

```
TITLE1 "Model 1b: Empty Means, Random Intercept Model";
PROC MIXED DATA=&datafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML;
  CLASS ID session;
  MODEL nm3rt = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=ID;
  REPEATED session / R TYPE=VC SUBJECT=ID; RUN;
```

In the PROC NLMIXED line below, adaptive Gauss-Hermite Quadrature (METHOD=GAUSS) is used to integrate over random effects (necessary for non-normal outcomes, but not really relevant here given that our random effects and residuals here are still assumed to be normally distributed.). Newton-Raphson optimization (TECH=NEWRAP) is a specific way of finding the top of the likelihood mountain. Finally, we also set stricter gradient convergence criteria (GCONV=1e-12) to ensure each parameter is really at the top of its dimension of the mountain.

```
TITLE1 "Model 1b: Empty Means, Random Intercept Model via NLMIXED";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;

* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances and covariances;
  PARS fint=1770
      VarU0=198820 VarE=44900;

* Setting up level-2 equations;
  b0i = fint + U0i;

* Setting up level-1 equation WITHOUT level-1 residual;
  PredY = (b0i);

* Telling it which DV, defining level-1 residual;
* RT is normally distributed with a mean of "PredY" and a variance of "VarE";
  MODEL nm3rt ~ normal(PredY, VarE);

* Random effects are normally distributed with means=0 and estimated variances;
  RANDOM U0i ~ normal([0],[VarU0]) SUBJECT=ID;

* Asking for ICC and SE;
  ESTIMATE "ICC" VarU0 / (VarU0 + VarE);
RUN;
```

MIXED OUTPUT:

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	ID	198820	29035	6.85	<.0001	ICC = .8158
Session	ID	44900	2825.63	15.89	<.0001	

Fit Statistics	
-2 Log Likelihood	8546.3
AIC (smaller is better)	8552.3
AICC (smaller is better)	8552.4
BIC (smaller is better)	8560.2

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1770.70	45.1952	101	39.18	<.0001

NLMIXED OUTPUT:

Fit Statistics	
-2 Log Likelihood	8546.3
AIC (smaller is better)	8552.3
AICC (smaller is better)	8552.4
BIC (smaller is better)	8560.2

Parameter Estimates									
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
fint	1770.69	45.1952	100	39.18	<.0001	0.05	1681.02	1860.36	-5.4E-6
VarU0	198820	29035	100	6.85	<.0001	0.05	141216	256424	2.82E-10
VarE	44900	2825.64	100	15.89	<.0001	0.05	39294	50506	3.324E-9

Additional Estimates								
Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
ICC	0.8158	0.02404	100	33.94	<.0001	0.05	0.7681	0.8635

Model 3b. Random Quadratic Time Model via MIXED and NLMIXED

```

TITLE1 "Model 3b: Random Quadratic Time Model";
PROC MIXED DATA=&datafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML;
CLASS ID session;
MODEL nm3rt = clsess clsess*clsess / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT clsess clsess*clsess / G V VCORR TYPE=UN SUBJECT=ID;
REPEATED session / R TYPE=VC SUBJECT=ID; RUN;

TITLE1 "Model 3b: Random Quadratic Time Model via NLMIXED";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second and third lines are variances and covariances;
PARMS fint=1946 flin=-121 fquad=14
VarU0=273306 CovU10=-35626 VarU1=25438
CovU20=3845 CovU21=-3838 VarU2=622 VarE=20298;
* Setting up level-2 equations;
b0i = fint + U0i;
b1i = flin + U1i;
b2i = fquad + U2i;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*clsess) + (b2i*clsess*clsess);
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL nm3rt ~ normal(PredY, VarE);
* Random effects are normally distributed with means=0 and estimated variances;
RANDOM U0i U1i U2i ~ normal([0,0,0],[VarU0,CovU10,VarU1,CovU20,CovU21,VarU2]) SUBJECT=ID;
RUN;
    
```

MIXED OUTPUT:

Estimated G Matrix					
Row	Effect	Person ID	Col1	Col2	Col3
1	Intercept	101	273306	-35262	3845.38
2	c1sess	101	-35262	25438	-3837.76
3	c1sess*c1sess	101	3845.38	-3837.76	622.81

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	273306	40828	6.69	<.0001
UN(2,1)	ID	-35262	11765	-3.00	0.0027
UN(2,2)	ID	25438	5781.19	4.40	<.0001
UN(3,1)	ID	3845.38	1920.35	2.00	0.0452
UN(3,2)	ID	-3837.76	968.79	-3.96	<.0001
UN(3,3)	ID	622.81	169.99	3.66	0.0001
Session	ID	20298	1649.11	12.31	<.0001

Fit Statistics	
-2 Log Likelihood	8321.8
AIC (smaller is better)	8341.8
AICC (smaller is better)	8342.1
BIC (smaller is better)	8367.9

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	1945.85	53.5825	101	36.32	<.0001
C1sess	-120.90	19.9481	101	-6.06	<.0001
C1sess*C1sess	13.8656	3.3985	101	4.08	<.0001

NLMIXED OUTPUT:

Iteration History					
Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1*	24	4160.89414	0.029504	0.003755	-0.05484
2*	36	4160.89137	0.002776	0.000271	-0.00332
3*	48	4160.88737	0.003992	0.000167	-0.00504
4*	60	4160.88433	0.003045	0.000053	-0.00511
5*	72	4160.8839	0.000433	4.276E-6	-0.00059

NOTE: ABSGCONV convergence criterion satisfied.

Fit Statistics	
-2 Log Likelihood	8321.8
AIC (smaller is better)	8341.8
AICC (smaller is better)	8342.1
BIC (smaller is better)	8367.9

Parameter Estimates									
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
fint	1945.85	53.5832	98	36.31	<.0001	0.05	1839.52	2052.18	1.81E-10
flin	-120.90	19.9571	98	-6.06	<.0001	0.05	-160.50	-81.2957	8.76E-11
fquad	13.8656	3.3987	98	4.08	<.0001	0.05	7.1210	20.6102	2.24E-10
VarU0	273313	40831	98	6.69	<.0001	0.05	192286	354341	-6.15E-7
CovU10	-35568	11799	98	-3.01	0.0033	0.05	-58982	-12154	-4.28E-6
VarU1	25474	5791.26	98	4.40	<.0001	0.05	13981	36966	-3.04E-6
CovU20	3898.39	1923.78	98	2.03	0.0454	0.05	80.7058	7716.07	1.039E-8
CovU21	-3841.28	969.79	98	-3.96	0.0001	0.05	-5765.79	-1916.77	-3E-7
VarU2	622.95	170.04	98	3.66	0.0004	0.05	285.52	960.38	-2.74E-6
VarE	20299	1649.21	98	12.31	<.0001	0.05	17026	23572	-8.05E-9

Because variances can be hard to estimate, the negative exponential models that follow instead estimate standard deviations, and then calculate variances by squaring them.

Model 6a. Negative Exponential Model (Fixed Asymptote, Fixed Amount, Fixed Rate)

```

TITLE1 "Negative Exponential Model via NLMIXED";
TITLE2 "Model 6a: Fixed Asymptote, Fixed Amount, Fixed Rate";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWWRAP GCONV=1e-12;

* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS   fasymp= 1600 famount=300 frate=-1
          SDE=600;

* Setting up level-2 equations;
b0i = fasymp;
b1i = famount;
b2i = frate;

* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*EXP(b2i*c1sless));

* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL nm3rt ~ normal(PredY, sdE*sdE);

* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote"   fasymp;
ESTIMATE "Fixed Amount"     famount;
ESTIMATE "Fixed Rate"       frate;
ESTIMATE "Residual E Variance" sdE*sdE;

* Creating extra parameters and predicted means;
ESTIMATE "Fixed Intercept"   fasymp+famount;
ESTIMATE "Session 1 Predicted Mean" fasymp+(famount*EXP(frate*0));
ESTIMATE "Session 2 Predicted Mean" fasymp+(famount*EXP(frate*1));
ESTIMATE "Session 3 Predicted Mean" fasymp+(famount*EXP(frate*2));
ESTIMATE "Session 4 Predicted Mean" fasymp+(famount*EXP(frate*3));
ESTIMATE "Session 5 Predicted Mean" fasymp+(famount*EXP(frate*4));
ESTIMATE "Session 6 Predicted Mean" fasymp+(famount*EXP(frate*5));

RUN;

```

Iteration History					
Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1*	15	4631.13397	7.008659	0.635591	-1452266
2*	21	4631.1316	0.002371	0.398959	-0.00545
3*	27	4631.1279	0.003703	0.349025	-0.00405
4*	35	4625.36573	5.762165	11.20453	-0.06333
5*	41	4624.07696	1.28877	1.987387	-1.51198
6*	48	4609.76677	14.3102	13.15472	-3.73214
7*	55	4609.06439	0.702377	6.387425	-7.52845
8*	61	4608.1577	0.906689	3.89459	-2.63132
9*	67	4607.16533	0.992366	0.236206	-1.46147
10*	73	4606.91758	0.247757	0.260529	-0.34753
11*	79	4606.77195	0.145622	0.951019	-0.1909
12	86	4606.69057	0.081389	1.227711	-0.61523
13	92	4606.60972	0.080842	1.38279	-0.21767
14	98	4606.57997	0.029751	0.310892	-0.05081
15	104	4606.57696	0.003013	0.044025	-0.00564
16	110	4606.57692	0.000043	0.000582	-0.00008
17	116	4606.57692	1.039E-8	1.704E-7	-2.08E-8

NOTE: AMSGCONV convergence criterion satisfied.

Fit Statistics	
-2 Log Likelihood	9213.2
AIC (smaller is better)	9221.2
AICC (smaller is better)	9221.2
BIC (smaller is better)	9238.8

Parameter Estimates									
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
fasymp	1675.25	54.8326	606	30.55	<.0001	0.05	1567.56	1782.93	1.069E-9
famount	284.71	64.5965	606	4.41	<.0001	0.05	157.85	411.57	3.15E-10
frate	-0.6698	0.4247	606	-1.58	0.1153	0.05	-1.5039	0.1643	1.704E-7
SDE	484.28	13.9107	606	34.81	<.0001	0.05	456.97	511.60	-336E-13

Additional Estimates									
Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	
Fixed Asymptote	1675.25	54.8326	606	30.55	<.0001	0.05	1567.56	1782.93	
Fixed Amount	284.71	64.5965	606	4.41	<.0001	0.05	157.85	411.57	
Fixed Rate	-0.6698	0.4247	606	-1.58	0.1153	0.05	-1.5039	0.1643	
Residual E Variance	234532	13474	606	17.41	<.0001	0.05	208071	260992	
Fixed Intercept	1959.96	47.8094	606	41.00	<.0001	0.05	1866.07	2053.85	
Session 1 Predicted Mean	1959.96	47.8094	606	41.00	<.0001	0.05	1866.07	2053.85	
Session 2 Predicted Mean	1820.97	36.2937	606	50.17	<.0001	0.05	1749.69	1892.24	
Session 3 Predicted Mean	1749.83	30.7816	606	56.85	<.0001	0.05	1689.38	1810.28	
Session 4 Predicted Mean	1713.42	23.4669	606	73.01	<.0001	0.05	1667.33	1759.51	
Session 5 Predicted Mean	1694.79	27.9049	606	60.73	<.0001	0.05	1639.98	1749.59	
Session 6 Predicted Mean	1685.25	36.3828	606	46.32	<.0001	0.05	1613.80	1756.70	

Model 6b. Negative Exponential Model (Random Asymptote, Fixed Amount, Fixed Rate)

```

TITLE1 "Negative Exponential Model via NLMIXED";
TITLE2 "Model 6b: Random Asymptote, Fixed Amount, Fixed Rate";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;

* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS fasymp= 1675 famount=284 frate=-.7
      sdE=474 sdU0=10;

* Setting up level-2 equations;
b0i = fasymp + U0i;
b1i = famount;
b2i = frate;

* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*EXP(b2i*clsess));

* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL nm3rt ~ normal(PredY, sdE*sdE);

* Defining random effects: normally distributed with means and variances;
RANDOM U0i ~ normal([0],[sdU0*sdU0]) SUBJECT=ID;

* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote"          fasymp;
ESTIMATE "Fixed Intercept"         fasymp+famount;
ESTIMATE "Fixed Amount"           famount;
ESTIMATE "Fixed Rate"             frate;
ESTIMATE "Residual E Variance"    sdE*sdE;
ESTIMATE "Random Asymptote U0 Variance" sdU0*sdU0;

* Creating extra parameters and predicted means;
ESTIMATE "Fixed Intercept"         fasymp+famount;
ESTIMATE "Session 1 Predicted Mean" fasymp+(famount*EXP(fraterate*0));
ESTIMATE "Session 2 Predicted Mean" fasymp+(famount*EXP(fraterate*1));
ESTIMATE "Session 3 Predicted Mean" fasymp+(famount*EXP(fraterate*2));
ESTIMATE "Session 4 Predicted Mean" fasymp+(famount*EXP(fraterate*3));
ESTIMATE "Session 5 Predicted Mean" fasymp+(famount*EXP(fraterate*4));
ESTIMATE "Session 6 Predicted Mean" fasymp+(famount*EXP(fraterate*5));

RUN;

```

Iteration History

Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1*	17	4284.53992	321.7295	2.366279	-111.108
2*	25	4260.22207	24.31785	0.670259	-67.5009
3*	37	4206.38115	53.84092	3.324296	-8.69485
4	44	4203.31763	3.063522	0.287907	-10.2693
5	51	4202.06928	1.248343	0.032384	-2.22851
6	58	4202.01577	0.053511	0.001224	-0.10347
7	65	4202.01562	0.00015	2.948E-6	-0.0003

NOTE: AMSGCONV convergence criterion satisfied.

Fit Statistics

-2 Log Likelihood	8404.0
AIC (smaller is better)	8414.0
AICC (smaller is better)	8414.1
BIC (smaller is better)	8427.1

Is the random asymptote variance significant?

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
fasymp	1675.25	49.2032	100	34.05	<.0001	0.05	1577.63	1772.87	3.31E-8
famount	284.71	24.5497	100	11.60	<.0001	0.05	236.00	333.41	1.102E-8
frate	-0.6698	0.1614	100	-4.15	<.0001	0.05	-0.9900	-0.3495	2.948E-6
sdE	184.05	5.7913	100	31.78	<.0001	0.05	172.56	195.54	-1.55E-7
sdU0	447.95	32.4064	100	13.82	<.0001	0.05	383.65	512.24	-1.59E-6

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
Fixed Asymptote	1675.25	49.2032	100	34.05	<.0001	0.05	1577.63	1772.87
Fixed Intercept	1959.96	48.1335	100	40.72	<.0001	0.05	1864.46	2055.45
Fixed Amount	284.71	24.5497	100	11.60	<.0001	0.05	236.00	333.41
Fixed Rate	-0.6698	0.1614	100	-4.15	<.0001	0.05	-0.9900	-0.3495
Residual E Variance	33875	2131.79	100	15.89	<.0001	0.05	29645	38104
Random Asymptote U0 Variance	200656	29033	100	6.91	<.0001	0.05	143056	258256
Fixed Intercept	1959.96	48.1335	100	40.72	<.0001	0.05	1864.46	2055.45
Session 1 Predicted Mean	1959.96	48.1335	100	40.72	<.0001	0.05	1864.46	2055.45
Session 2 Predicted Mean	1820.97	46.6578	100	39.03	<.0001	0.05	1728.40	1913.53
Session 3 Predicted Mean	1749.83	46.0819	100	37.97	<.0001	0.05	1658.40	1841.25
Session 4 Predicted Mean	1713.42	45.4558	100	37.69	<.0001	0.05	1623.24	1803.60
Session 5 Predicted Mean	1694.79	45.8166	100	36.99	<.0001	0.05	1603.89	1785.68
Session 6 Predicted Mean	1685.25	46.6678	100	36.11	<.0001	0.05	1592.66	1777.84

Re-estimated with:

PARMS fasymp= 1675 famount= 284 frate= -.7
sdE= 184 sdU0=447;

The estimates can be very sensitive to start values, so try different values to make sure the estimates don't change!

Model 6c. Negative Exponential Model (Random Asymptote, Random Amount, Fixed Rate)

```

TITLE1 "Negative Exponential Model via NL MIXED";
TITLE2 "Model 6c: Random Asymptote, Random Amount, Fixed Rate";
PROC NL MIXED DATA=&datafile. METHOD=GAUSS TECH=NEWRAP GCONV=1e-12;

* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS fasymp=1675 famount=284 frate=-.7
      sdE=184 sdU0=447 sdU01=1 sdU1=10;

* Setting up level-2 equations;
b0i = fasymp + U0i;
b1i = famount + U1i;
b2i = frate;

* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*EXP(b2i*c1sless));

* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "y" and a variance of "VarE";
MODEL nm3rt ~ normal(PredY, sdE*sdE);

* Defining random effects: normally distributed with means and variances;
RANDOM U0i U1i ~ normal([0,0],[sdU0*sdU0,sdU01*sdU01,sdU1*sdU1]) SUBJECT=ID;

* Labeling estimated parameters;
ESTIMATE "Fixed Asymptote"          fasymp;
ESTIMATE "Fixed Amount"            famount;
ESTIMATE "Fixed Rate"              frate;
ESTIMATE "Residual E Variance"     sdE*sdE;
ESTIMATE "Random Asymptote U0 Variance" sdU0*sdU0;
ESTIMATE "Asymptote-Amount U01 Covariance" sdU01*sdU01;
ESTIMATE "Random Amount U1 Variance" sdU1*sdU1;
ESTIMATE "Asymptote-Amount Correlation" (sdU01*sdU01)/(sdU0*sdU1);

* Creating extra parameters and predicted means;
ESTIMATE "Fixed Intercept"          fasymp+famount;
ESTIMATE "Session 1 Predicted Mean" fasymp+(famount*EXP(frate*0));
ESTIMATE "Session 2 Predicted Mean" fasymp+(famount*EXP(frate*1));
ESTIMATE "Session 3 Predicted Mean" fasymp+(famount*EXP(frate*2));
ESTIMATE "Session 4 Predicted Mean" fasymp+(famount*EXP(frate*3));
ESTIMATE "Session 5 Predicted Mean" fasymp+(famount*EXP(frate*4));
ESTIMATE "Session 6 Predicted Mean" fasymp+(famount*EXP(frate*5));
RUN;

```

Re-estimated with: **PARMS** fasymp= 1675 famount= 284 frate= -.7
 sdE= 152 sdU0=437 sdU01= 82 sdU1= 277;

Iteration History					
Iter	Calls	NegLogLike	Diff	MaxGrad	Slope
1*	20	4174.64313	27.25232	2.965592	-3.6011
2*	29	4167.35401	7.289123	0.533215	-27.975
3*	38	4163.88955	3.46446	0.105733	-6.22334
4*	47	4163.76983	0.119717	0.009226	-0.23289
5*	56	4163.76888	0.000951	0.000415	-0.00107
6*	70	4163.70031	0.068574	0.232078	-0.00046
7*	79	4163.67024	0.030063	0.005606	-0.05804
8*	88	4163.67021	0.000031	5.535E-6	-0.00006

NOTE: AMSGCONV convergence criterion satisfied.

Fit Statistics	
-2 Log Likelihood	8327.3
AIC (smaller is better)	8341.3
AICC (smaller is better)	8341.5
BIC (smaller is better)	8359.6

Is the random amount variance significant?

Parameter Estimates

Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper	Gradient
fasymp	1683.48	45.4523	99	37.04	<.0001	0.05	1593.30	1773.67	-3.75E-6
famount	279.94	33.5457	99	8.35	<.0001	0.05	213.38	346.51	-1.59E-7
frate	-0.7533	0.1181	99	-6.38	<.0001	0.05	-0.9877	-0.5189	5.535E-6
sdE	151.79	5.3422	99	28.41	<.0001	0.05	141.19	162.39	-2.98E-7
sdU0	436.83	31.8997	99	13.69	<.0001	0.05	373.54	500.13	-6.18E-7
sdU01	81.5157	90.6707	99	0.90	0.3708	0.05	-98.3947	261.43	-3.39E-6
sdU1	277.95	28.4727	99	9.76	<.0001	0.05	221.45	334.44	-1.86E-7

Additional Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Lower	Upper
Fixed Asymptote	1683.48	45.4523	99	37.04	<.0001	0.05	1593.30	1773.67
Fixed Amount	279.94	33.5457	99	8.35	<.0001	0.05	213.38	346.51
Fixed Rate	-0.7533	0.1181	99	-6.38	<.0001	0.05	-0.9877	-0.5189
Residual E Variance	23039	1621.75	99	14.21	<.0001	0.05	19821	26257
Random Asymptote U0 Variance	190823	27870	99	6.85	<.0001	0.05	135523	246122
Asymptote-Amount U01 Covariance	6644.80	14782	99	0.45	0.6540	0.05	-22686	35976
Random Amount U1 Variance	77254	15828	99	4.88	<.0001	0.05	45848	108659
Asymptote-Amount Correlation	0.05473	0.1224	99	0.45	0.6557	0.05	-0.1881	0.2975
Fixed Intercept	1963.43	54.7439	99	35.87	<.0001	0.05	1854.80	2072.05
Session 1 Predicted Mean	1963.43	54.7439	99	35.87	<.0001	0.05	1854.80	2072.05
Session 2 Predicted Mean	1815.29	47.2407	99	38.43	<.0001	0.05	1721.55	1909.03
Session 3 Predicted Mean	1745.54	45.0393	99	38.76	<.0001	0.05	1656.17	1834.91
Session 4 Predicted Mean	1712.70	44.3143	99	38.65	<.0001	0.05	1624.77	1800.63
Session 5 Predicted Mean	1697.24	44.3919	99	38.23	<.0001	0.05	1609.16	1785.32
Session 6 Predicted Mean	1689.96	44.7051	99	37.80	<.0001	0.05	1601.26	1778.67

$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm (1.96 * \sqrt{\text{Random Variance}})$$

$$\text{Asymptote 95\% CI} = \gamma_{00} \pm (1.96 * \sqrt{\tau_{U_0}^2}) \rightarrow 1,683.5 \pm (1.96 * \sqrt{190,823}) = 827 \text{ to } 2,540$$

$$\text{Amount 95\% CI} = \gamma_{10} \pm (1.96 * \sqrt{\tau_{U_1}^2}) \rightarrow 279.9 \pm (1.96 * \sqrt{77,254}) = -265 \text{ to } 825$$

Model 6d. Negative Exponential Model (Random Asymptote, Random Amount, Random Rate)

```
TITLE1 "Negative Exponential Model via NLMIXED";
TITLE2 "Model 6d: Random Asymptote, Random Amount, Random Rate";
PROC NLMIXED DATA=&datafile. METHOD=GAUSS TECH=NEWWRAP GCONV=1e-12;
* Must define all parameters to be estimated and provide start values;
* First line is fixed effects, second line is variances;
PARMS fasymp=1678.25 famount=-282.72 frate=-.7323
      sdE=130.88 sdU0=426.79 sdU01=87.68 sdU1=290.57
      sdU02=5.1343 sdU12=.08921 sdU2=1.3483;
* Setting up level-2 equations;
b0i = fasymp + U0i;
b1i = famount + U1i;
b2i = frate + U2i;
* Setting up level-1 equation WITHOUT level-1 residual;
PredY = (b0i) + (b1i*EXP(b2i*c1sess));
* Telling it which DV, defining level-1 residual;
* RTs is normally distributed with a mean of "PredY" and a variance of "VarE";
MODEL nm3rt ~ normal(PredY , sdE*sdE);
* Defining random effects: normally distributed with means and variances;
RANDOM U0i U1i U2i ~ normal([0,0,0],[sdU0*sdU0,sdU01*sdU01,sdU1*sdU1,
sdU02*sdU02,sdU12*sdU12,sdU2*sdU2]) SUBJECT=ID; RUN;
```



From the log:

ERROR: Quadrature accuracy of 0.000100 could not be achieved with 31 points. The achieved accuracy was 1.000000.

No convergence, after several tries with different start values and relaxing the estimation options....

```
* Simpler estimation to get start values;
PROC NLMIXED DATA=&datafile. METHOD=FIRO;
```


So how did we do? Let's compare model predictions in terms of means (top) and variances (bottom)?



So which is the best unconditional model of within-person change for these RT data? To compare across the polynomial, piecewise, and exponential families, we need to use ML to estimate each.

Model	Total # Parameters	ML -2LL	ML AIC	ML BIC
1b Most Parsimonious Baseline: Empty Means, Random Intercept	3	8546.3	8552.3	8560.2
6c Negative Exponential: Random Asymptote, Random Amount, Fixed Rate	7	8327.3	8341.3	8359.6
3b Polynomial: Random Quadratic Time	10	8321.8	8341.8	8367.9
4c Piecewise: Random Slope12, Random Slope26	10	8298.9	8318.9	8345.1
0 Least Parsimonious Baseline: Saturated Means, Unstructured Variance	27	8278.1	8332.1	8402.7
<hr/>				
1b Fit better than Empty Means, Random Intercept Model?	Δdf	ML -2 ΔLL	$p <$	
6c Negative Exponential: Random Asymptote, Random Amount, Fixed Rate	4	219.0	.001	
3b Polynomial: Random Quadratic Time	7	224.5	.001	
4c Piecewise: Random Slope12, Random Slope26	7	247.4	.001	
0 Least Parsimonious Baseline: Saturated Means, Unstructured Variance	24	268.2	.001	
<hr/>				
0 Fit worse than Saturated Means, Unstructured Variance Model?	Δdf	ML -2 ΔLL	$p <$	
1b Most Parsimonious Baseline: Empty Means, Random Intercept	24	268.2	.001	
6c Negative Exponential: Random Asymptote, Random Amount, Fixed Rate	20	49.2	.001	
3b Polynomial: Random Quadratic Time	17	43.7	.001	
4c Piecewise: Random Slope12, Random Slope26	17	20.8	.235	

Based on making theoretical sense, I'd say negative exponential wins, but based purely on these empirical data, piecewise slopes wins.